
1 Inledning

Denna text beskriver (p̊a ett ganska löst och översiktligt sätt medges) och diskuterar
exempel p̊a en inkrementell konstruktionsstrategi. I detta fall var det osäkert om, eller
hur bra, det hela skulle fungera till slut s̊a det var inte möjligt att skriva en initial
kravspecifikation. Texten tar fasta p̊a att man lär sig mer om ett problem efter att ha
st̊angats mot det ett tag.

Erfarenheten om h̊ardvaran och problemet bakas sedan under utvecklingen in i konstruk-
tionen:1

Figur 1.1: All ingenjörsmässig problemlösning handlar om stegvis förfining fr̊an en given,
än s̊a länge bristfällig, funktion.

Texten är skriven i möjligen lite slapp, icke-formell stil. Det är avsiktligt och min för-
hoppning är att det ger lite inblick i att allt inte är klart genomtänkt fr̊an början. Det
fanns en idé som jag ville testa, men i övrigt gick det mesta p̊a fritid och ganska lösa
boliner.

1Figuren fr̊an https://hackaday.com/2017/11/30/life-on-contract-how-much-do-i-charge/

2 Digitalt filter p̊a ATmega328

För att undersöka om det g̊ar att utföra digital signalbehandling p̊a en 8-bitars AVR-
processor vid blott 16 MHz lödde jag ihop ett Arduino-kort med lite enkel kringelektronik.

Det handlade om ett ing̊angssteg med lite förstärkning, en snabbare digitalomvandlare
än vad ATmega328p har internt, ett efterföljande l̊agpassfilter och en audioförstärkare
för att kunna koppla in hörlurar eller högtalare till utsignalen, se figuren nedan.

Först.  
Arduino 16 MHz

(ATmega328p)
D/A LP PA

Ljud in

Figur 2.1: Överst visas signalkedjan fr̊an insignal till högtalare. Signalen omvandlas till
digitala värden i den interna A/D-omvandlaren i ATmega328p, behandlas i
programvara innan de digitala värdena p̊aförs D/A-omvandlare (D/A), l̊ag-
passfiltreras (LP) och slutligen förstärks (PA). Underst visas realiseringen i
form av lödda komponenter p̊a kopparlaminat. Utsignalen är kraftig nog att
höras (ljudligt!) i en ansluten högtalare.

3 Koppla, löd och provkör delarna

Det exakta elektriska schemat är ointressant för oss här. Det viktiga är att beskriva hur
uppkopplingen av h̊ardvara skedde p̊a ett strukturerat sätt, steg för steg med täta prov
av funktionerna:

• I detta fall lödde jag ihop ing̊angssteget först (längst till höger i figur 2.1) och
kontrollerade att att förstärkningen och DC-offset kunde ändras med tv̊a poten-
tiometrar som avsett. Förstärkningspotentiometern ersattes slutligen med ett fast
motst̊and.

• Sedan monterades Arduinokortet med smältlim och D/A-omvandlaren löddes p̊a
plats. Den D/A-omvandlare som används styrs med serieprotokollet SPI s̊a testkod
för att skicka ut n̊agot p̊a SPI skrevs och att n̊agot hände p̊a avsedda utg̊angar
mättes med oscilloskop. Exakt vad som hände brydde jag mig inte om utan räknade
kallt med att om rätt saker kom till D/A-omvandlaren skulle n̊agot hända p̊a dess
utg̊ang.1

• Efter detta kopplades SPI till D/A-omvandlaren. Men en oscilloskopprob p̊a om-
vandlarens utsignal och en programloop som skickade ut en uppräknad byte fr̊an
$00 till $FF förväntade jag mig en spänningsramp p̊a oscilloskopet. Efter lite data-
bladsläsning b̊ade för processorn och MPC4802 blev det just precis s̊a.

• Sedan kopplades de rent analoga resterande delarna upp ocks̊a. Med signalgenerator
p̊afördes efter hand sinusv̊agor av olika frekvens för att konstatera att LP-filtret
fungerade som avsett med en brytfrekvens i storleken n̊agra kilohertz och att lju-
det hördes i högtalaren. Mot det ritade schemat var enda ändringen att signalen
behövde dämpas ordentligt innan volymkontrollen. Lite experimenterande gav att
ett 68k motst̊and i serie med LP-filtrets utg̊ang blev bra. Lägg märke till detta
tillvägag̊angssätt: Koppla lite, prova att det blev som tänkt, åtgärda eventuellt,
koppla lite, prova att det blev som tänkt, åtgärda eventuellt, koppla lite, prova att
det blev som tänkt, åtgärda eventuellt, koppla lite, prova att det blev som tänkt,
åtgärda eventuellt, koppla lite, . . . Hela tiden sm̊a steg.

Vid det här laget visste jag att hela kedjan av h̊ardvara fungerade som avsett!

1Detta är ett rätt stort steg och motiveras av erfarenhet. Hade jag var mer noggrann hade jag försökt
identifiera att det som hände p̊a SPI-utg̊angarna verkligen var det som programmet skickade. För
att kunna se det p̊a ett oscilloskop skulle programmet behöva skicka exakt samma byte om och om
igen. Ingen orimligt steg, men jag hoppade över det den här g̊angen. Väl medveten om att faktiskt
behöva göra det (suck!) om det strulade senare.

4 F̊a ihop n̊agot som funkar

Innan n̊agra egentliga signalbehandlingsrutiner kunde skrivas behövs ett antal sm̊a ru-
tiner för att hantera h̊ardvaran. Det handlar om att konfigurera in- och utg̊angar, SPI,
AD-omvandlare och interna timers.

D/A-omvandlarrutin D/A-omvandlaren är en MCP48021 vilken kräver SPI-kommuni-
kation. Lyckligtvis inneh̊aller ATmega328p h̊ardvara för SPI s̊a för denna del behövs
n̊agra I/O-register konfigureras. Som nämnts m̊aste MCP4802:s datablad konsulteras
för att reda ut exakt vad som skall skickas till den och ett program m̊aste skrivas för att
testa att den är korrekt inkopplad och att vi kan styra den. Det första programmets syfte
var att enbart resultera i en subrutin för att skriva till omvandlaren. Efter lite hyfsning
nöjde jag mig med nedanst̊aende kod som en frist̊aende subrutin:

; --- DAC MCP48x2 send r16 (0..FF)

;

dac:

push r17

swap r16

mov r17 ,r16

andi r17 ,$0F

ori r17 ,$10

cbi PORTB ,2 ; SS on device low

out SPDR ,r17 ; send high byte

dac1: in r17 ,SPSR ; get status

sbrs r17 ,SPIF ; wait

jmp dac1

out SPDR ,r16 ; send low byte

dac2: in r17 ,SPSR ; get status

sbrs r17 ,SPIF ; wait

jmp dac2

sbi PORTB ,2 ; SS on device high

pop r17

ret

Den uppmärksamma läsaren noterar att n̊agra rader återkommer här. S̊a här tidigt i
utvecklingen är jag dock nöjd om det fungerar även om det inte är optimalt. I detta sig-

1https://www.electrokit.com/mcp4802e-p.51070

4 F̊a ihop n̊agot som funkar

nalbehandlingsfall är det väsentligt med snabb kod s̊a jag är inte benägen att onödigtvis
slöa ner rutinens körtid.

Avbrott Hela samplingsmekanismen är h̊ardvarustyrd. En intern timer (tmr0) använ-
des för att ge periodiska avbrott med en frekvens av 62.52 kHz / 8 = 7.812 kHz och starta
en AD-omvandling. När omvandlingen är klar (efter 13 klockpulser av AD-klockans fre-
kvens) triggas ett adcConversionComplete-avbrott. Det är i denna avbrottsrutin som
det samplade värdet hämtas, eventuellt behandlas och skickas ut till D/A-omvandlaren.

För att, enligt rubriken, först f̊a n̊agot som funkar ihop, utfördes denna avbrottsrutin
p̊a enklaste sätt, det vill sägaden enbart reläade inlästa samplade värden direkt till DA-
omvandlaren:

; --- ADC conversion complete

;

adcConversionComplete:

push r16

lds r16 ,SREG

push r16

lds r16 ,ADCH ; r16 = 0..256

call dac ; output channel

pop r16

sts SREG ,r16

pop r16

reti

Notera: Processorn ATmega328p har en del av sina I/O-register p̊a ställen som inte
kan n̊as med in- och out-instruktionerna. Därför används lds och sts som ersättning
i denna processor. ATmega16 är väluppfostrad i detta sammanhang s̊a där kan man
använda de vanliga I/O-instruktionerna.

Nu hade jag n̊agot som samplade ljudet fr̊an min iPhone och skickade ut det. Hädanefter
vill jag enbart tillföra förfiningar som inte förstör, efter varje liten ändring laddade jag
ner koden till processorn för att konstatera att inget var trasigt.

216 Mhz / 256 = 62.5 kHz

4 F̊a ihop n̊agot som funkar

En första modifiering var att markera för mig själv var signalbehandlingskoden skulle
ligga. Samtidigt ville jag inte fr̊anhända mig möjligheten att återigen kontrollera om
h̊ardvaran fungerade som avsett, s̊a labeln passThru: lades till:

; --- ADC conversion complete

;

adcConversionComplete:

push r16

lds r16 ,SREG

push r16

lds r16 ,ADCH ; r16 = 0..256

jmp passThru

;

; space for processing code

;

passThru:

call dac ; output channel

pop r16

sts SREG ,r16

pop r16

reti

Hädanefter kan jag välja att köra signalbehandlingsdelen eller inte genom att kommente-
ra bort raden jmp passThru. Det är överraskande praktiskt att ha s̊adana kortslutningar
av i koden för att snabbt kunna konstatera om h̊ardvaran fortfarande fungerar. Här var
det oftast för att konstatera att sladden mellan ing̊angssteget och arduinokortet lossnat
igen. . .

5 Förfina

Digital signalbehandling är inte speciellt förl̊atande: Man HÖR när n̊agot g̊att snett!
Tro mig. . . Skickar man ut fel byte av misstag, eller om beräkningarna tar för l̊ang tid
och avbrottsrutinen avbryter vid fel tid och s̊a vidare, hörs det omedelbart. S̊a resten av
programutveckligen skedde nästan hela tiden med musik i högtalaren som tecken p̊a att
senaste versionen av programmet fungerar. Vilket är trevligt p̊a mer än ett sätt.

FIR-filter En lämplig förstaövning för att bedöma processorns lämplighet i signalbe-
handlingssammanhang är att implementera ett FIR-filter. Ett FIR är i princip en medel-
värdesbildare av flera samplade invärden. Ett ”2-tap” FIR-filter leverar för varje nytt
inkommet sample medelvärdet av detta och det förra samplet. Här implementerades
först ett 5-tap-filter:

T  

senaste
sample 

T T T

äldsta
sample 

+  

utsignal

h4h0 h1 h2 h3

Figur 5.1: Filtrets utsignal beräknas för varje inkommit nytt sample. Om konstanterna
hi är lika med 1 genom antalet tap (här 1

5) utför FIR-filtret en ren medelvärdes-
bildning, en sorts l̊agpassfiltrering. Med fouriertransformering kan koefficien-
terna beräknas för andra filterfunktioner. Med snabba signalprocessorer är
flera hundra tappar möjliga.

Samtliga samplevärden är 8-bitars tv̊akomplementskodade tal och ligger i en lista i
SRAM mellan en startadress $0200 och slutadress $0200 + Buff_Size.

Koden för filtret traverserar listan, multiplicerar de lagrade samplevärdena med h0, h1, . . .
enligt figuren samt summerar dem i r25:r24.

5 Förfina

Bra program använder pekare. Det ger tidigt en stor flexibilitet och är används pekare
fr̊an början: XH:HL pekar ut rätt sample i bufferten och ZH:ZL pekar ut korrekt multi-
plikator ur tabellen fir5Tab. Avslutningsvis lagras höga delen av resultatet i SRAM i
positionen firOut.

;

; --- fir5 5-tap FIR -filter

;

; lastSample(x4)*h0+butlast(x3)*h1+..+ oldest(x0)*h4

; build in r25 :24.

; store result in firOut (8bit)

fir5:

push r18

push XL

clr r24 ; sum = 0

clr r25

ldi ZH,HIGH(fir5Tab *2)

ldi ZL,LOW(fir5Tab *2)

ldi XL,Buff_Size ; adress of last (one beyond lastSample)

fir51:

ld r16 ,X ; lastSample

lpm r18 ,Z+

muls r16 ,r18 ; r1:r0

add r24 ,r0

adc r25 ,r1

dec XL

brpl fir51

;lsl r25

sts firOut ,r25

pop XL

pop r18

ret

fir5Tab:

.db 31,31,31,31,31

Med denna kod genomfördes ett antal tester för att studera filtrets effektivitet men
framför allt för att utröna om det är tillräckligt med 8-bitars värden b̊ade som samples
och konstanter. Och detta hela tiden med hänsyn till tids̊atg̊ang.

Prov med 10-bitars sample gav att de extra bitarna inte verkade spela n̊agon roll —
antingen försvann de i beräkningarna pga avrundningsfel eller s̊a p̊averkade de änd̊a inte
den 8-bitars D/A-omvandlingen — s̊a det bestämdes att h̊alla kvar 8-bitars sample.

Däremot verkade det (och inte helt entydigt här) som att 16-bitars värden var av godo
för filterkoefficienterna. Att använda b̊ade 10-bitars sample och 16-bitars koefficienter
bedömdes ta för l̊ang beräkningstid.

5 Förfina

En modifierad kod för de beslutade 8-bitars sample och 16-bitars konstanter blev till
slut:

fir5:

push r18

push XL

clr r22 ; sum = 0

clr r23

clr r24

ldi ZH,HIGH(firTab *2)

ldi ZL,LOW(firTab *2)

ldi XL,Buff_Size ; adress of last (one beyond lastSample)

fir51:

ld r16 ,X ; lastSample 2’s compl

lpm r18 ,Z+ ; r19:r18 16bit 2’s compl constant

lpm r19 ,Z+

lsl r18 ; extra gain

mulsu r16 ,r18 ; Multiply signed with signed

add r22 ,r0

adc r23 ,r1

adc r24 ,zero

mul r16 ,r19 ; Multiply unsigned

add r23 ,r0 ; Add without carry

adc r24 ,zero ; collect carry

dec XL

brpl fir1

;lsl r25

sts firOut ,r23

pop XL

pop r18

ret

fir5Tab:

.dw 125 ,125 ,125 ,125 ,125 ,125

Där additionen sker löpande i ett 24-bits register som utgörs av r24:r23:r22. Jag är
osäker p̊a om r24 n̊agonsin används, det handlar om storleken p̊a fir5Tab-värdena, men
f̊ar st̊a kvar tills vidare.

Bli bekant med problemet Med koden i detta skick utfördes sedan åtskilliga prov med
upp till 31 tappars filter enbart för att jag skulle bekanta mig med koden och systemet.
Det var lätt att höra vad som verkligen spelade roll och vad som var mindre viktigt.

För att inte f̊a för ”kornigt” ljud i högtalaren var det till exempel viktigt att ha s̊a stora
variabelvärden som möjligt i alla beräkningar, dock utan att f̊a overflow vid n̊agot enstaka

5 Förfina

tillfälle, för d̊a knäpper det obehagligt starkt i högtalaren. Med för sm̊a konstanter å
andra sidan blir de (konstanterna) alla ungefär lika stora vilket gör dels att alla filter
liknar varann och dels att blir utsignalen svag (ibland s̊a svag att den blir 0, det vill säga
tysta partier försvinner helt!).

Three Cycle Semi-Log

15 tap FIR no window
15 tap FIR triangular window
15 tab FIR Bandpass @ fs/4

fs=62.5/8=7812 Hz
Prototype: ...0,1,0...

7812/16=488 Hz

F (dB)

f (Hz)

BW band pass

I
488

Figur 5.2: Uppmätt frekvensg̊ang hos 15-taps FIR-filter med utsignalen normerad till
0 dB. Filtret är designat efter ett prototypfilter med s̊a smal bandbredd som
möjligt (488 Hz). De faktiska filterkoefficienterna beräknades genom FFT och
skalades s̊a att overflow inte uppstod i beräkningarna. Den röda kurvan visar
frekvensg̊angen med omodifierade konstanter, den streckade med konstanter
viktade med ett triangelfönster. Den bl̊a kurvan är resultatet efter att konstan-
terna multiplicerats med +1, 0,−1, 0,+1, . . . vilket enligt teorin skall förflytta
hela filtret till samplingsfrekvensen/4, fs/4

Rensa kod och fastställ omgivningarna Under hela denna viktiga experimentperiod
ändrades kommentarerna i koden allteftersom s̊a att det alltid betydde rätt sak. Onö-
diga push/pop togs bort. Överhuvudtaget blev jag efter hand mer och mer bekant med
koden och teorin bakom signalbehandlingen. Det känns skönt att teorin bekräftas av
experimenten (med undantag för bandpassfiltret ovan. Jag undrar fortfarande om det
blev helt rätt?).

5 Förfina

Det kändes efter ett tag jobbigt att ha alla varianter av testade filter: fir5, fir7, fir15
och fir31 i koden varför en enda fir-rutin implementerades.

Initialt hade jag felaktigt förutsett att jag skulle behöva nollställa samplebufferten, plocka
ut enstaka värden ur densamma och utföra en 16x16→32 bitars multiplikation. De rutiner
jag skrivit för detta rensades ut. Annan död kod togs bort och det hela snyggades till s̊a
att det blev mer läsbart.

D̊a jag fr̊an början var osäker p̊a vilken samplingstakt som var möjlig (egentligen tids̊at-
g̊angen för beräkningar mellan samplen) provades samplingstakterna fs = 62.5 kHz och
fs = 62.5/8 = 7.812 kHz. Den senare valdes.

Experiment undertogs ocks̊a med olika hastigheter p̊a ADC-omvandlingens klockfrekvens.
Databladet anger 200 kHz som högsta frekvens men öppnar även lite diffust för högre
frekvenser om försämrad precision kan till̊atas. Det visade sig att 1 MHz fungerar utan
hörbar försämring av ljudet.

I och med dessa avslutande designbeslut för man betrakta denna första ”proof of concept”-
fas som över. Nu finns en grund att bygga vidare p̊a där man vet att

• den analoga h̊ardvaran fungerar, kanske inte ner till LSB i D/A-omvandlingen, men
tillräckligt bra för att fortsätta utveckla p̊a. Ett korrekt designat mönsterkort ger
säkert bättre signaler ifr̊aga om brum och brus och därmed ljudkvalitet. Det blir i
vart fall inte sämre.

• mikrokontrollerns AD-omvandlare och interna timrar fungerar p̊a ett tillfredsstäl-
lande sätt, SPI likas̊a.

• med en samplingsperiod p̊a 1/7812 = 128µs hinns en hel del beräkningar med.
Olika uppmätta ungefärliga tider under projektets g̊ang äro:

Åtgärd Tids̊atg̊ang µs

dac 5
A/D-omvandling 13
FIR 5 tap 9
FIR 7 tap 12
FIR 15 tap 22
FIR 31 tap 42

Macrot

.macro bit3

sbi PORTD ,3

cbi PORTD ,3

.endmacro

användes flitigt som skvallersignal för att bestämma dessa tider.

Hädanefter handlar allt om att skriva kod!

6 Bygg p̊a funktionalitet!

Bortsett fr̊an att implementera filter enligt ovan kan man utföra mycket intressant sig-
nalbehandling om insignalen kan delas upp tv̊a ortogonala delar, en in-phase- och en
out-of-phase-signal.

Den senare kallas ofta quadrature-phase varför de brukar förkortas till signalerna I och
Q. De kan d̊a ses som vektorer ~v = I + jQ i det komplexa talplanet vilket avsevärt
underlättar till exempel studier av signalens faslägen vilket i sig öppnar för avkodning
av fasmodulerade signaler.

Längden av ~v, |v| =
√
I2 + Q2, kan beräknas efter kvadrering och rotutdragning eller

med cordic1-algoritmen.

v 
Q 

I 
Re 

Im 

Figur 6.1: Insignalen ~v kan delas upp i I- och Q-komposanter.

De rektangulära koordinaterna kan D/A-omvandlas och kopplas till ett xy-kopplat oscil-
loskop för vidare studium. Den använda omvandlaren MCP4802 har lyckligtvis tv̊a ut-
kanaler varför h̊ardvaran praktiskt taget är förberedd för detta redan.

Kvadratursampling Med kvadratursampling menas att sampla med en frekvens som
är fyra g̊anger högre än den önskade bandbredden. Med fs = 7.8 kHz kan teoretiskt
signaler med frekvensinneh̊all upp till fs/4 ≈ 2 kHz behandlas.

Metoden är en form av korrelering med en cosinus- och en sinusv̊ag i de punkter dessa
är 0 eller ±1. I praktiken väljer man ut vilka samples som skall adderas (+1), subtra-
heras (−1) eller inte ing̊a (0) i signalerna I och Q. Detta tillvägag̊angssätt medför att

1https://en.wikipedia.org/wiki/CORDIC

6 Bygg p̊a funktionalitet!

dyra multiplikationer inte behöver genomföras. Däremot m̊aste de korrelerade samplen
filtreras/medelvärdesbildas innan de används vidare.

De b̊ada signalerna I- och Q utgör rektangulära koordinater i det komplexa talplanet.

Grafiskt beskrivs kvadratursampling i figuren nedan.

A/D  

* 

cos(2pi n/4)= 1, 0, –1, 0, ...  

FIR I

* FIR Q

–sin(2pi n/4)= 0, –1, 0, 1, ...  

Figur 6.2: Principschema över kvadratursampling. Samplingen utförs med 4x den öns-
kade bandbredden och respektive kvadratursignal filteras innan de behandlas
vidare.

Kodmodifieringar Med stegvis förfining som princip m̊aste koden förses med tv̊a buffrar
(en för I- och en för Q-samplen), tv̊a utsignaler (en för vardera filtrerade I- och Q-samplen)
och aktivering av den hittills outnyttjade andra D/A-omvandlarkanalen.

fir Rutinen modifieras att användas mot valfri buffert. Anropet m̊aste ange mot vilken
buffer filtreringen skall ske och vart resultatet skall g̊a. Här visas koden innan respektive
efter de inkrementella förändringarna.

Konstanter och minnesupplägg. Före:

.equ FIR_SIZE = 5 ; <-- set this!

; --- dont fiddle below. Automatic dependencies!

.equ BUFF_SIZE = FIR_SIZE - 1

.dseg

.org SRAM_START

firOut: .byte 1

.org $200

Buff:

.byte BUFF_SIZE

lastSample:

.byte 1 ; last sample from A/D

; must be at XH:BUFF_SIZE

6 Bygg p̊a funktionalitet!

Konstanter och minnesupplägg. Efter:

.equ FIR_SIZE = 5 ; <-- set this!

; --- dont fiddle below. Automatic dependencies!

.equ BUFF_SIZE = FIR_SIZE - 1

.dseg

.org SRAM_START

OutI: .byte 1

OutQ: .byte 1

; IN-PHASE buffer

.org $200

IBuff: .byte BUFF_SIZE

lastSampleI:

.byte 1 ; last sample from A/D

; must be at XH:BUFF_SIZE

; QUADRATURE -PHASE buffer

.org $300

QBuff: .byte BUFF_SIZE

lastSampleQ:

.byte 1 ; last sample from A/D

Som synes huvudsakligen en dubblering av buffrar och namngivning av minnesplatser
*I och *Q.

6 Bygg p̊a funktionalitet!

Själva filterkoden döptes om till det mer korrekta generiska namnet fir och rensades
upp n̊agot. För att korta ner framställningen nedan har oväsentlig kod ersatts med ”:”.

FIR-filterkod. Före:

fir5:

push r18

push XL

clr r22 ; sum = 0

clr r23

clr r24

ldi ZH,HIGH(firTab *2)

ldi ZL,LOW(firTab *2)

ldi XL,Buff_Size ; adress of last (one beyond lastSample)

fir51:

ld r16 ,X ; lastSample 2’s compl

lpm r18 ,Z+ ; r19:r18 16bit 2’s compl constant

lpm r19 ,Z+

:

brpl fir1

sts firOut ,r23

pop XL

pop r18

ret

fir5Tab:

.dw 125 ,125 ,125 ,125 ,125

FIR-filterkod. Efter:

fir:

clr r22 ; sum = 0

clr r23

clr r24

ldi ZH,HIGH(firTab *2)

ldi ZL,LOW(firTab *2)

ldi XL,Buff_Size ; adress of last (one beyond lastSample)

fir1:

ld r16 ,X ; lastSample 2’s compl

lpm r18 ,Z+ ; r19:r18 16bit 2’s compl constant

lpm r19 ,Z+

:

brpl fir1

mov r16 ,r23

ret

firTab:

.dw 125 ,125 ,125 ,125 ,125

r16 inneh̊aller nu det beräknade returvärdet.

6 Bygg p̊a funktionalitet!

Anrop av den senare rutinen blir nu n̊agon av nedanst̊aende

; --- filter IN -PHASE

sts lastSampleI ,r16 ; lastSample contains newest sample

ldi XH,HIGH(IBuff) ; which buffer to use

call fir ; ALWAYS 8-bit result

call pushBuff ; push lastSample

sts OutI ,r16 ; 8 bit -128..+127

; --- filter QUADRATURE -PHASE

sts lastSampleQ ,r16 ; lastSample contains newest sample

ldi XH,HIGH(QBuff) ; which buffer to use

call fir ; ALWAYS 8-bit result

call pushBuff ; push lastSample

sts OutQ ,r16 ; 8 bit -128..+127

pushBuff är en rutin som skiftar värdena i den utpekade bufferten ner̊at ett steg och
lägger lastSample p̊a buffertens nyaste plats. pushBuff agerar p̊a den buffer XH för
tillfället pekar ut.

För att komma till dessa anrop och retur av resultatet genomfördes ett antal sm̊a steg
vilka var för sig inte gjorde koden okörbar:

1. Genomfördes att fir kunde arbeta mot den buffer som angavs (IBuff eller QBuff).

2. Genomfördes att pushBuff kunde arbeta mot den buffer som angavs (IBuff eller
QBuff).

3. Genomfördes att lastSampleI användes istället för lastSample.

4. Genomfördes att resultatet lämnades i r16 och sts OutI,r16 lades till efter an-
ropet.

5. Den fungerande IN-PHASE-koden kopierades till QUADRATURE-PHASE-koden. last-
SampleI böts mot lastSampleQ, samt OutI mot OutQ.

6. De b̊ada kodstyckena provades sedan var för sig för att avslutningsvis konstatera
att b̊ada fungerar som avsett.

Notera att koden kompilerades och provkördes efter varje punkt ovan. P̊a detta sätt
kunde jag vara säker p̊a att koden hela tiden fungerade.

Och, viktigast, om den plötsligt inte fungerat skulle det vara lätt att g̊a tillbaka och ändra
till fungerande kod igen; felet uppstod ju i den senaste modifieringen. Felsökningen2

eliminieras praktiskt taget med denna metod med sm̊a ändringar mellan fungerande
programversioner.

Slutsatsen är att försöka genomföra s̊a försiktiga kodmodifieringar att koden
hela tiden är i ett fungerande skick!

2I betydelsen att man g̊ar igenom koden och letar efter möjliga fel.

6 Bygg p̊a funktionalitet!

Vad som menas med försiktiga modifieringar varierar fr̊an programmerare till program-
merare men behöver inte vara mer än en rad om det känns bekvämast och är möjligt
genomförbart.

För D/A-omvandlingen behövdes en rutin för omvandlarens kanal A och en för kanal B.
Min försiktiga modifiering av dac ovan blev tv̊a ing̊angar, dacA och dacB, i huvudsakligen
samma rutin:

; --- dacA/B MCP48x2 send r16 (0..FF)

dacA:

swap r16

mov r17 ,r16

andi r17 ,$0F

ori r17 ,$10 ;$10=daca

jmp dac0

dacB:

swap r16

mov r17 ,r16

andi r17 ,$0F

ori r17 ,$90 ; $90=dacb

dac0:

cbi PORTB ,2 ; SS on device low

out SPDR ,r17 ; send high byte

dac1: in r17 ,SPSR ; get status

sbrs r17 ,SPIF ; wait

jmp dac1

out SPDR ,r16 ; send low byte

dac2: in r17 ,SPSR ; get status

sbrs r17 ,SPIF ; wait

jmp dac2

sbi PORTB ,2 ; SS on device high

ret

Det kanske g̊ar att komprimera denna rutin mer men d̊a p̊a bekostnad av ökad exe-
kveringstid. Den avvägningen lämnas till senare.

6 Bygg p̊a funktionalitet!

Sammanfattning Rubrikerna hittills sammanfattar de olika stegen i en
ingenjörsmässig lösningsg̊ang:

1. Bestäm, åtminstone i grova drag, det slutliga systemets funktion.

Undvik här att tänka att tänka p̊a hur uppgiften skall lösas, tänk mer p̊a det
önskade slutresultatet utseende och funktion. Detta steg kan vara en ensidig skiss
eller en mer komplett kravspecifikation.

2. Koppla, löd och provkör delarna.

Utan fungerande delsystem kommer det sammansatta kompletta systemet aldrig
kunna fungera. Delsystemen m̊aste konstrueras och testas s̊a de utför de önskade
funktionerna.

3. F̊a ihop n̊agot som funkar överhuvudtaget.

Använd de just konstruerade delsystemen som byggklossar för att bygga ett fun-
gerande system enligt punkt 1.

4. Förfina.

Här ägnar man rätt mycket tid att konstatera att funktionen verkligen uppfyller
kraven. Med fungerande kod kan man faktorisera, modifiera och förbättra funktio-
nen. Ofta utan att funktionen n̊agon g̊ang upphör! Modifiera koden för generalitet
och stabilitet.

5. Bygg p̊a funktionalitet.

Med det fungerande och stabila systemet som grundpl̊at kan man nu stegvis bygga
vidare för utökad funktionalitet.

Alla stegen 1–4 ovan sker i försiktiga och sm̊a steg. N̊agot som kanske inte framkommit
tydligt är vikten av att kunna mäta resultat ofta, ofta. Efter varje liten kodförändring
mättes resultat upp. I detta fall testades ocks̊a koden kontinuerligt i och med att ljud
hela tiden spelades upp i högtalaren. För övrigt användes oscilloskopet precis hela tiden,
det startades vid varje arbetspass och användes till passets slut.

